NAG Library Routine Document

G01HCF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G01HCF returns probabilities for the bivariate Student's t-distribution, via the routine name.

2 Specification

```
FUNCTION G01HCF (TAIL, A, B, DF, RHO, IFAIL)
REAL (KIND=nag_wp) G01HCF
INTEGER            DF, IFAIL
REAL (KIND=nag_wp) A(2), B(2), RHO
CHARACTER(1)       TAIL
```

3 Description

Let the vector random variable $X = (X_1, X_2)^{\mathrm{T}}$ follow a bivariate Student's t-distribution with degrees of freedom ν and correlation ρ, then the probability density function is given by

$$f(X:\nu,\rho) = \frac{1}{2\pi\sqrt{1-\rho^2}}\left(1+\frac{X_1^2+X_2^2-2\rho X_1 X_2}{\nu\left(1-\rho^2\right)}\right)^{-\nu/2-1}.$$

The lower tail probability is defined by:

$$P(X_1 \leq b_1, X_2 \leq b_2 : \nu, \rho) = \int_{-\infty}^{b_1}\int_{-\infty}^{b_2} f(X:\nu,\rho)dX_2dX_1.$$

The upper tail probability is defined by:

$$P(X_1 \geq a_1, X_2 \geq a_2 : \nu, \rho) = \int_{a_1}^{\infty}\int_{a_2}^{\infty} f(X:\nu,\rho)dX_2dX_1.$$

The central probability is defined by:

$$P(a_1 \leq X_1 \leq b_1, a_2 \leq X_2 \leq b_2 : \nu, \rho) = \int_{a_1}^{b_1}\int_{a_2}^{b_2} f(X:\nu,\rho)dX_2dX_1.$$

Calculations use the Dunnet and Sobel (1954) method, as described by Genz (2004).

4 References

Dunnet C W and Sobel M (1954) A bivariate generalization of Student's t-distribution, with tables for certain special cases *Biometrika* **41** 153–169

Genz A (2004) Numerical computation of rectangular bivariate and trivariate Normal and t probabilities *Statistics and Computing* **14** 151–160

5 Parameters

1: TAIL – CHARACTER(1) *Input*

On entry: indicates which probability is to be returned.

TAIL = 'L'
The lower tail probability is returned.

TAIL = 'U'
The upper tail probability is returned.

TAIL = 'C'
The central probability is returned.

Constraint: TAIL = 'L', 'U' or 'C'.

2: A(2) – REAL (KIND=nag_wp) array *Input*

On entry: if TAIL = 'C' or 'U', the lower bounds a_1 and a_2.

If TAIL = 'L', A is not referenced.

3: B(2) – REAL (KIND=nag_wp) array *Input*

On entry: if TAIL = 'C' or 'L', the upper bounds b_1 and b_2.

If TAIL = 'U', B is not referenced.

Constraint: if TAIL = 'C', $a_i < b_i$, for $i = 1, 2$.

4: DF – INTEGER *Input*

On entry: ν, the degrees of freedom of the bivariate Student's t-distribution.

Constraint: DF ≥ 1.

5: RHO – REAL (KIND=nag_wp) *Input*

On entry: ρ, the correlation of the bivariate Student's t-distribution.

Constraint: $-1.0 \leq$ RHO ≤ 1.0.

6: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

If on exit, IFAIL $\neq 0$, then G01HCF returns zero.

IFAIL $= 1$

On entry, TAIL is not valid.

IFAIL $= 3$

On entry, $\mathrm{B}(i) \leq \mathrm{A}(i)$ for central probability, for some $i = 1, 2$.

IFAIL $= 4$

On entry, DF < 1.

IFAIL $= 5$

On entry, RHO < -1.0 or RHO > 1.0.

7 Accuracy

Accuracy of the algorithm implemented here is discussed in comparison with algorithms based on a generalized Placket formula by Genz (2004), who recommends the Dunnet and Sobel method. This implementation should give a maximum absolute error of the order of 10^{-16}.

8 Further Comments

None.

9 Example

This example calculates the bivariate Student's t probability given the choice of tail and degrees of freedom, correlation and bounds.

9.1 Program Text

```
    Program g01hcfe

!     G01HCF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: g01hcf, nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: prob, rho
      Integer                          :: df, ierr, ifail
      Character (1)                    :: tail
!     .. Local Arrays ..
      Real (Kind=nag_wp)               :: a(2), b(2)
!     .. Executable Statements ..
      Write (nout,*) 'G01HCF Example Program Results'
      Write (nout,*)

!     Skip heading in data file
      Read (nin,*)

!     Display titles
      Write (nout,*) &
        ' A(1)         B(1)       A(2)        B(2)        DF  RHO     TAIL P'
      Write (nout,*)

d_lp: Do
        ierr = 0
        a(1:2) = 0.0_nag_wp
        b(1:2) = 0.0_nag_wp

        Read (nin,Fmt='(a1)',Advance='no',Iostat=ierr) tail
```

```
!       Read parameter values
        Select Case (tail)
        Case ('l','L')
          Read (nin,*,Iostat=ierr) df, rho, b(1), b(2)
        Case ('c','C')
          Read (nin,*,Iostat=ierr) df, rho, a(1), b(1), a(2), b(2)
        Case ('u','U')
          Read (nin,*,Iostat=ierr) df, rho, a(1), a(2)
        Case Default
          Write (nout,*) 'Invalid problem specification in data file'
          Exit d_lp
        End Select

        If (ierr/=0) Then
          Exit d_lp
        End If

!       Calculate probablity
        ifail = 0
        prob = g01hcf(tail,a,b,df,rho,ifail)

!       Display results
        Select Case (tail)
        Case ('l','L')
          Write (nout,99999,Advance='no') '-Inf', b(1), '-Inf', b(2)
        Case ('u','U')
          Write (nout,99998,Advance='no') a(1), 'Inf', a(2), 'Inf'
        Case ('c','C')
          Write (nout,99997,Advance='no') a(1), b(1), a(2), b(2)
        End Select

        Write (nout,99996) df, rho, tail, prob
      End Do d_lp

99999 Format (1X,2(A4,8X,E11.4,1X))
99998 Format (1X,2(E11.4,2X,A3,8X))
99997 Format (1X,4(E11.4,1X))
99996 Format (I3,1X,F7.4,2X,A1,2X,F8.4)
    End Program g01hcfe
```

9.2 Program Data

```
G01HCF Example Program Data
L 8 0.6 4.0 0.8                 :LOWER_TAIL DF RHO B(1) B(2)
C 12 -0.2 -40.0 2.0 0.0 4.0     :CENTRAL DF RHO A(1) B(1) A(2) B(2)
U 2 0.3 -2.0 8.0                :UPPER_TAIL DF RHO A(1) A(2)
```

9.3 Program Results

```
 G01HCF Example Program Results

  A(1)         B(1)       A(2)        B(2)        DF  RHO     TAIL P

 -Inf         0.4000E+01 -Inf         0.8000E+00   8  0.6000  L    0.7764
 -0.4000E+02  0.2000E+01  0.0000E+00  0.4000E+01  12 -0.2000  C    0.4876
 -0.2000E+01  Inf         0.8000E+01  Inf          2  0.3000  U    0.0059
```